Modeling Non-Ideal Inductors in SPICE

Martin O'Hara Technical Manager, Newport Components, U.K. November 8 1993

Abstract

The non-ideal inductor exhibits both resonance and non-linear current characteristics. These effects can be modelled in SPICE by adding only 3 additional elements to model the real inductor characteristics of dc resistance, wire capacitance and magnetic core loss. The values for these model parameters can all be obtained from standard data sheet parameters via a few simple calculations. The resulting model gives accurate impedance and phase simulations over a wide frequency range and over the peak resonance frequency. The dc current saturation characteristics is modelled by a simple 2nd order polynomial that gives a close simulation to measured performance over 2.5 times the recommended dc current limit. Comparisons of measured inductor performance and simulation results are given to illustrate the proximity of the models to real inductor behaviour.

Introduction

Modelling of inductors and inductive elements in SPICE has always been of low importance to analogue designers. This is partly because SPICE was developed primarily for IC design where inductive elements are usually parasitic and very small.

The widespread use of SPICE for discrete analogue circuit design has seen the program being used to analyze switching power supplies and filters in which the behaviour of the inductive element is critical to the accuracy of the simulation. In general these circuits operate using ideal inductors reasonably well since the current and frequency of operation are in the ideal operating region of the inductors used (i.e. relatively low frequency and well below the saturation current limit).

More recent applications employing inductive elements are electromagnetic interference (EMI) filters, in which the resonances across a very wide range of frequencies needs to be examined. Likewise employing inductors in dc supply filters can put the inductor near its dc saturation region. In both later cases the modelling of the non-ideal behaviour of the inductor is important for accurate predictions of circuit performance.

Real Inductor Behaviour

In an ideal inductor the impedance (Z) is purely reactive and proportional to the inductance (L) only; The phase of signal across the ideal inductor would always be +90° out of phase with the applied voltage and there would be no effect of DC current bias on the behaviour of an ideal inductor.



Figure 1

If we compare the measured frequency response for the impedance of a real inductor to the ideal model we can see two distinct differences at either end of the frequency spectrum (figure 1). At the low frequency (near DC) there is a dominant resistive element, observed in the constant impedance value and loss of the phase shift. At high frequency the inductor goes though a resonance peak and the impedance then falls and a voltage phase shift of -90° is observed, indicative of capacitive dominance. The frequency response is therefore observed to be non-ideal, however, it can be stated that near ideal behaviour does occur over the majority of the inductors operating region.



Figure 2

Under DC current bias there is a loss of inductance due to magnetic saturation. This is observed as a fall in inductance as the DC current through the inductor is increased.

Modelling Non-Ideal Behaviour

There are essentially two non-ideal characteristics[1] that are encountered when using an inductor; one is the resonance of the inductor and the other is magnetic saturation. Since these essentially act in different analyses in SPICE (i.e. AC or DC analyses), they can be considered separately, although combined into a single model.



Figure 3: Basic Inductor Model

The additional parasitics that cause the behaviour of an inductor to be non-ideal over the frequency range can be easily visualised and characterised. There are essentially two additional parameters that contribute; the dc resistance of the wire and its self capacitance (figure 3). These two additional parameters can usually be easily obtained from the specification for the inductor, hence additional measurement by the circuit designer should not be required, just a few simple calculations.

The series resistance is obtained simply from the quoted dc resistance of the inductor (R_{dc}). The parallel capacitance(C_p) can be obtained from the self resonant frequency of the inductor, since at this frequency the reactance of the wire capacitance (X_C) and the reactance of the inductance (X_L) are equal. Hence the capacitance can be expressed as;

$$C_p = \frac{1}{(2\pi f_o)^2 L_o} \qquad \textbf{(1)}$$

Where f_o is the self resonant frequency.

The effect of dc current causing magnetic saturation can be modelled as a simple second order polynomial. In SPICE 2G6 this was available directly in the standard polynomial inductor model by using the POLY key word after the node description.

The polynomial is specified by the equation;

$$L_I = L_o + L_1 I + L_2 I^2 + \ldots + L_n I^n \qquad \textbf{(2)}$$

where n≤20.

An inductor specification usually gives the dc current (I_{dc}) at which the inductance falls to 90% of its nominal value (L_o). Hence, using a second order approximation, the equation becomes;

$$0.9\, L_o = L_o + L_2 I_{dc}^2 \qquad \textbf{(3)}$$

Yielding a second order coefficient of;

$$L_2 = -\frac{0.1\, L_o}{I_{dc}^2} \qquad \textbf{(4)}$$

Note that the first order co-efficient will have to be specified as zero.

In SPICE 3E2 the polynomial inductor is no longer available and a more complex method of modelling this effect is required using the non-linear element B and a zero value voltage source to measure the current through the inductor.

The complete polynomial inductor of SPICE 2G6 can be written as a subcircuit in SPICE 3E2.

XL 1 2 POLYL L_o L_1 L_2 L_3 ....

.SUBCKT POLYL 1 2
V1 1 3 DC 0
LO 3 2 L_o
B1 2 3 I=I(V1)^2*L_1/(2*L_o)+I(V1)^3*
+ L_2/(3*L_o)+I(V1)^4*L_3/(4*L_o)+....
.ENDS

Here we are only interested in modelling the second order polynomial, hence only the L_2 term is of interest. This can be determined from the SPICE 2G6 coefficients, or directly from the maximum DC current value.

$$L_{3E2} = \frac{L_2}{3\, L_o} = -\frac{1}{30\, I_{dc}^2} \qquad \textbf{(5)}$$

The polynomial sub-circuit can hence be rewritten;

.SUBCKT POLYL 1 2
V1 1 3 DC 0
LO 3 2 L_o
B1 2 3 I=I(V1)^3*L_{3E2}
.ENDS

It should be noted that this subcircuit only replicates the polynomial equation from version 2G6, the additional model elements also need to be added.

Simulation and Test Results

A radial leaded bobbin inductor (14 105 40) was measured for the non-ideal characteristics described above. Impedance and phase were determined on a Hewlett-Packard HP4192A Low Frequency Impedance Analyzer, dc current characteristic was determined on a Wayne-Kerr (WK) 3245 Precision Impedance Analyzer and 3220 Bias Unit.

The effect of dc current saturation proved long winded to simulate. The reason for this is that an AC analysis cannot be performed concurrently with a DC sweep. Hence the dc current through the inductor had to be manually changed and the circuit re-simulated to get a simulation of impedance over a range of dc current (the inductance was calculated from the simulated impedance characteristic from SPICE and read directly from the WK3245).



Figure 2: Impedance Analysis

Discussion

Impedance results proved to be exceptionally well matched, the only discrepancy being a slight difference in the resonant frequency. The difference in resonant frequency is purely a production variation, the model is centred on the typical value of 800kHz, whereas the sample used was resonant at 696kHz.

In simulation it is important that the measuring instrument is modelled as closely as possible so that any effects this may load onto the component is determined. The problem of the measurement system model is clearly illustrated in the phase results. If a 50O oscillator source to load impedance is used (as suggested in the HP manual) a poor simulated phase response is observed due to the loading of the source, however, using a 1MO impedance gave an accurate simulated match to the measurement characteristic (the simulated impedance response was the same with either source impedance).



Figure 3: DC Current Analysis

The simulated dc current characteristic looks dissimilar to the measured result. The initial inductance is higher for the measured part (1.05mH) and it can be observed that the characteristic is more likely a 3rd order polynomial expression. However, the simulation is reasonably close and estimates a worse case (particularly since the I_{dc} current value for the sample used was nearer 5A; Newport inductors are always specified conservatively). The shape of the characteristic is reasonably close over 2.5 times the parts recommended operating current shown and using the 2nd order polynomial rather than a 3rd means that additional measurements are not required. What the simulation implies is that it results in the worse case characteristic for the least effort.



Figure 4: Peak Resonance Impedance Analysis

Peak Resonance

If the peak resonance is more closely examined it is observed that there is some disparity between simulation[2] and measurement for both the peak impedance result (figure 4) and rate of phase change (figure 5). This can be expected in that there is no provision is this simple model for the finite loss in the magnetic material.



Figure 5: Peak Resonance Phase Analysis

The magnetic loss can be modelled reasonably well as a parallel resistor (R_p) across the existing model. The value can again be calculated from data sheet parameters, using the quality factor (Q). In a parallel RLC[3] circuit the relationship between the quality factor and inductance is given by;

$$Q = \frac{R_p}{2\pi\, f_o L_o} \quad \textbf{(6)}$$

The data sheet value for the 14 105 40 inductor used here is Q=49, hence a parallel resistance of 246kO is calculated (225kO using the actual values for the sample part).

The resulting simulations for impedance and phase now match the measured results exceptionally well. This indicates the method for determining R_p is a reasonable approximation from a circuit designers point of view.

The above simulation results suggest the model gives a reasonably good approximation to the real behaviour of the inductor over a wide frequency and current range. The improvement has also been gained for no additional measurements, which means that the model can be derived from the component specification.

Limitations

The above model is now quite sophisticated for an inductive element, however, there are still limitations and this should be borne in mind. The model assumes that there is no variance of resistance and capacitance with dc current, at low values of these parameters this may be adequate as these will tend to be swamped by the rest of the circuit.

Negative inductance values can be obtained when the dc current exceeds approximately 3.16 times the I_{DC} value[1]. In the SPICE 3E2 sub-circuit this can be compensated for by putting in an IF..THEN conditional statement, this can either set the value of the inductor to zero, or some specified value. In SPICE 2G6 this facility is not available, it is therefore advisable to have a some measure of the dc current in the circuit element if it is suspected that the dc current is greater than 2.5 times the I_{DC} value.

Parameter Tolerances

The tolerance for inductance is usually specified in the data sheet (±10% for the sample used), however, few of the other parameters have a tolerance figure. In the cases of R_{dc} and I_{dc} these are worse case values and no other tolerance is required.

The tolerance of the self resonant frequency is related to the inductance value and wire capacitance. The tolerance of the wire capacitance is difficult to estimate accurately since, even with machine wound products, the value is so small that slight variations in winding cause noticeable changes in the capacitance. As an estimate, it could be expected that ±20% variation in the value of C_p would be observed.

The tolerance in the R_p value is also difficult to determine, even the core manufacturers do not usually specify tolerances of the loss parameters of the core. Again a ±20% tolerance is predicted to be sufficient to allow a Monte-Carlo analysis to accurately predict worse cases.



Figure 6: Completed Inductor Model

Summary

It is possible to simulate several complex aspects of inductor operation in SPICE using only 3 extra passive elements (figure 6) and a simple polynomial expression. The resulting model gives accurate inductor simulations over a wide range of operating conditions with a minimal increase in computation time (only one extra node is introduced) and no additional measurements are required.

SPICE 2G6 Example

The following example is a model for a Newport Components 1400 series 1mH inductor (14 105 40) where L_o=1mH, R_{dc}=0.173O, I_{dc}=4.0A, Q=49 and f_o=800kHz.

```
.SUBCKT IND14105  1  2
LO  3  2  POLY  1E-3  0  -6.25E-6
RDC 1  3  0.173
CP  1  2  39.6E-12
RP  1  2  250K
.ENDS
```

SPICE 3E2 Example

The following example is a model for a Newport Components 1400 series 1mH inductor (14 105 40) where L_o=1mH, R_{dc}=0.173O, I_{dc}=4.0A, Q=49 and f_o=800kHz.

```
.SUBCKT L14105  1  2
LO  1  4  1E-3
V1  4  3  DC 0
B1  4  1  I=I(V1)^3*-2.08E-3
RDC 3  2  0.173
CP  1  2  39.6E-12
RP  1  2  250K
.ENDS L14105
```

References

[1] SPICE Version 2G1 User's Guide, A. Vladimirescu, A.R. Newton, D.O. Pederson, UCB, 1980 (+2G6 upgrade, undated).

[2] Electric Circuits, 2nd ed., J.A. Edminster, Schaum's Outline Series, McGraw-Hill, 1983.

[3] Soft Ferrites, 2nd ed., E.C. Snelling, Butterworths, 1988.

[4] IsSPICE 3 User's Guide, Intusoft, 1992

[5] Private communication, M. Penberth, Technology Sources Ltd, November 1993.

[6] Inductor Databook, Newport Components Limited, 1993.

Newport Components Limited, 4 Tanners Drive,
Blakelands North, Milton Keynes, MK14 5NA
Tel: 01908 615232 Fax: 01908 617545